フリーのソフト Maxima を使って原子軌道と分子軌道を計算してみよう

Maxima のダウンロード

Maxima の使い方は例えば ここ を見てください

原子軌道

原子に束縛された1個の電子の状態



古典力学

軌道面を (x, y) 面にとる



$$m\frac{d^2x}{dt^2} = -\frac{Zq^2}{4\pi\varepsilon_0 r^2}\frac{x}{r}$$

$$m\frac{d^2y}{dt^2} = -\frac{Zq^2}{4\pi\varepsilon_0 r^2}\frac{y}{r}$$

$$\left.\begin{aligned} x &= r\cos\phi \\ y &= r\sin\phi \\ \alpha &= \frac{Zq^2}{4\pi\varepsilon_0 m} \end{aligned}\right\}$$ と置くと

遠心力　　クーロン力

$$\frac{d^2r}{dt^2} = r\left(\frac{d\phi}{dt}\right)^2 - \frac{\alpha}{r^2}$$

$$mr^2\frac{d\phi}{dt} = l$$

角運動量の保存

$$\vec{l} = \vec{r}\times\vec{p}$$

古典力学

$$\frac{d^2 r}{dt^2} = r\left(\frac{d\phi}{dt}\right)^2 - \frac{\alpha}{r^2}$$

$$mr^2 \frac{d\phi}{dt} = l$$

$$\longrightarrow \quad \frac{d^2 r}{dt^2} = \frac{\beta}{r^3} - \frac{\alpha}{r^2}$$

$$\alpha = \frac{Zq^2}{4\pi\varepsilon_0 m}$$

$$\beta = \frac{l^2}{m^2}$$

$r = \dfrac{1}{s}$ と置くと

$$\frac{dr}{dt} = -\frac{1}{s^2}\frac{ds}{dt} = -\frac{1}{s^2}\frac{ds}{d\phi}\frac{d\phi}{dt} = -\sqrt{\beta}\frac{ds}{d\phi}$$

$$\frac{d^2 r}{dt^2} = -\sqrt{\beta}\frac{d^2 s}{d\phi^2}\frac{d\phi}{dt} = -\beta s^2 \frac{d^2 s}{d\phi^2}$$

$$\longrightarrow \quad \frac{d^2 s}{d\phi^2} = -s + \frac{\alpha}{\beta}$$

$$\longrightarrow \quad s = \frac{\alpha}{\beta}\left[1 + e\cos\phi\right]$$

$a = \dfrac{\beta}{\alpha}\dfrac{1}{1-e^2}$ と置くと

$$r = \frac{a\left(1-e^2\right)}{1+e\cos\phi}$$

r は $a(1\text{-}e)$ と $a(1+e)$ の間の値をとる

古典力学



原子核の位置は楕円の焦点
e は離心率
a は長半径
b は短半径

楕円軌道

$$x=a(\cos u-e)$$

$$y=b\sin u$$

$$t=\frac{u-e\sin u}{\omega}$$

$$\omega=\sqrt{\frac{Zq^2}{4\pi\varepsilon_0 m}\frac{1}{a^3}}$$

保存量

エネルギー $E=-\dfrac{Zq^2}{8\pi\varepsilon_0}\dfrac{1}{a}$

角運動量 $\vec{l}=\vec{r}\times\vec{p}$

$$l=\sqrt{\frac{mZq^2}{4\pi\varepsilon_0 a}}\,b$$

軌道を等時間間隔でプロットする

```
m:500;
t:makelist(i*2*%pi/m,i,0,m);
load(newton);
findu(t):=newton(x-e*sin(x)-t,t);
a:1;
e:0.9;
u:makelist(findu(float(i*2.0*%pi/m)),i,0,m);
x:makelist(a*cos(u[i])-e*a,i,1,m+1);
y:makelist(sqrt(1-e^2)*a*sin(u[i]),i,1,m+1);
plot2d([discrete,x,y],[style,points]);
```



電子と原子核の距離を時間を横軸にしてプロットする

```
r:makelist(a*(1-e*cos(u[i])),i,1,m+1);
plot2d([discrete,t,r],[style,points]);
```



電子の速度をプロットすると真円になる

```
vx:makelist(-a*sin(u[i])/(1-e*cos(u[i])),i,1,m+1);
vy:makelist(sqrt(1-e^2)*a*cos(u[i])/(1-e*cos(u[i])),i,1,m+1);
plot2d([discrete,vx,vy],[style,points]);
```



量子論　半古典的な扱い



$$\frac{mv^2}{r} = \frac{Zq^2}{4\pi\varepsilon_0 r^2}$$

$$E = \frac{mv^2}{2} - \frac{Zq^2}{4\pi\varepsilon_0 r}$$

$\Rightarrow$

$$r = \frac{Zq^2}{4\pi\varepsilon_0 mv^2}$$

$$E = -\frac{Zq^2}{8\pi\varepsilon_0 r}$$

電子は波



円周に沿って定在波が立つ条件

$$2\pi r = \lambda n \qquad \lambda : \text{波長}$$

ド･ブロイの関係

$$\lambda = \frac{h}{mv}$$

$\Rightarrow$

$$r = \frac{\varepsilon_0 h^2}{\pi m Z q^2} n^2 = a_0 n^2$$

$$a_0 = \frac{\varepsilon_0 h^2}{\pi m Z q^2}$$

Bohr 半径≒0.529Å
($Z = 1$)

$$E = -\frac{mZ^2 q^4}{8\varepsilon_0^2 h^2} \frac{1}{n^2} = -\frac{Z^2 Ry}{n^2}$$

$$Ry = \frac{mq^4}{8\varepsilon_0^2 h^2}$$

Rydberg≒13.6eV

量子力学的扱い



電子の状態は波動関数で表わされる

電子の波動関数　Ψ

電子の存在確率　$|\Psi|^2$

Schrödinger方程式

$$-\frac{\hbar^2}{2m}\left(\frac{\partial^2\Psi}{\partial x^2}+\frac{\partial^2\Psi}{\partial y^2}+\frac{\partial^2\Psi}{\partial z^2}\right)-\frac{Zq^2}{4\pi\varepsilon_0 r}\Psi=E\Psi$$

運動エネルギー　　Coulomb ポテンシャル

極座標



$$x = r\sin\theta\cos\phi$$
$$y = r\sin\theta\sin\phi$$
$$z = r\cos\theta$$

$$-\frac{\hbar^2}{2m}\left[\frac{1}{r}\frac{\partial^2}{\partial r^2}(r\Psi)+\frac{1}{r^2}\left\{\frac{1}{\sin\theta}\frac{\partial}{\partial\theta}\left(\sin\theta\frac{\partial\Psi}{\partial\theta}\right)+\frac{1}{\sin^2\theta}\frac{\partial^2\Psi}{\partial\phi^2}\right\}\right]-\frac{Zq^2}{4\pi\varepsilon_0 r}\Psi = E\Psi$$

固有関数 $\Psi_{nlm} = R_{nl}(r)Y_{lm}(\theta,\phi)$ ← 球調和関数(Spherical harmonics)

角運動量 $\vec{l} = \vec{r}\times\vec{p}$

$$\frac{1}{\hbar^2}\vec{l}\cdot\vec{l}\ Y_{lm} = -\frac{1}{\sin\theta}\frac{\partial}{\partial\theta}\left(\sin\theta\frac{\partial Y_{lm}}{\partial\theta}\right)-\frac{1}{\sin^2\theta}\frac{\partial^2 Y_{lm}}{\partial\phi^2} = l(l+1)Y_{lm}$$

$$l_z Y_{lm} = \hbar m Y_{lm}$$

Laguerre 関数

$$R_{nl}(r) = \left(\frac{2}{a_0 n}\right)^{2/3}\left[\frac{(n-l-1)!}{2n(n+l)!}\right]^{1/2}\left(\frac{2r}{a_0 n}\right)^{l} e^{-r/a_0 n} L_{n-l-1}^{2l+1}\left(\frac{2r}{a_0 n}\right)$$

固有エネルギー $E_{nlm} = -\frac{Z^2 m q^4}{32\pi^2\varepsilon_0\hbar^2}\cdot\frac{1}{n^2}$

n = 1, 2, 3, …..
l = 0, 1, …, n-1
m = l, l-1, …, -l+1, -l

原子軌道を表す4つの量子数



主量子数	n = 1, 2, 3, ….	(K殻、L殻、…)
軌道量子数	l = 0, 1, 2, …., n –1	
磁気量子数	m = –l, … –1, 0, 1, … , l	
スピン量子数	s = –1/2, 1/2	

l	軌道	
0	s	*sharp*
1	p	*principal*
2	d	*diffuse*
3	f	*fundamental*
4	g	
5	h	

エネルギーは主量子数 n のみで決まる $E = -\dfrac{Z^2 Ry}{n^2}$

原子軌道　$n\ l$ で表わす　1s, 2s, 2p, 3s, 3p, 3d, 4s, 4p, 4d, 4f, ...

古典運動との対応

保存量

エネルギー $E = -\dfrac{Zq^2}{8\pi\varepsilon_0}\dfrac{1}{a} \longrightarrow a = n^2 a_B$ ⟺ 主量子数 n : 長半径の量子化

角運動量 $L = \sqrt{\dfrac{mZq^2}{4\pi\varepsilon_0 a}}\, b \longrightarrow b = nla_B$ ⟺ 軌道量子数 l: 短半径の量子化



同一エネルギーでの軌道



動径方向の波動関数 $R_{nl}(r)$ をプロットする。

```
load("orthopoly");
assume(%a0>0);
Rnl(n,l,r) := sqrt((2/(%a0*n))^3*(n-l-1)!/(2*n*((n+l)!)))*(2*r/(%a0*n))^l*exp(
 -r/(%a0*n))* gen_laguerre(n-l-1,2*l+1,2*r/(%a0*n));
Ylm(l,m,theta,phi) := spherical_harmonic(l,m,theta,phi);
psi(n,l,m,r,theta,phi) := Rnl(n,l,r)*Ylm(l,m,theta,phi);
%a0:1;
plot2d([Rnl(1,0,r),Rnl(2,0,r),Rnl(3,0,r)],[r,0,50],[y,-0.1,0.2]);
plot2d([Rnl(2,1,r),Rnl(3,1,r),Rnl(4,1,r)],[r,0,50],[y,-0.1,0.2]);
plot2d([Rnl(3,2,r),Rnl(4,2,r),Rnl(5,2,r)],[r,0,50],[y,-0.1,0.2]);
```

s ($l = 0$) $n = 1, 2, 3$



p ($l = 1$) $n = 2, 3, 4$



d ($l = 2$) $n = 3, 4, 5$



s ($l = 0$) で原子核の位置 ($r = 0$) で波動関数の値が大きい
l が大きくなるほど、電子は原子核から遠いところで存在確率が大きくなる
波動関数の節 ($\Psi = 0$) の数は $n - l - 1$ 個

動径方向の電子の存在確率 $r^2 R_{nl}(r)^2$

```
plot2d([r^2*Rnl(1,0,r)^2,r^2*Rnl(2,0,r)^2,r^2*Rnl(3,0,r)^2],[r,0,50]);
plot2d([r^2*Rnl(2,1,r)^2,r^2*Rnl(3,1,r)^2,r^2*Rnl(4,1,r)^2],[r,0,50]);
plot2d([r^2*Rnl(3,2,r)^2,r^2*Rnl(4,2,r)^2,r^2*Rnl(5,2,r)^2],[r,0,50]);
```

$s\ (\ l = 0\ )\ n = 1, 2, 3$



$p\ (\ l = 1\ )\ n = 2, 3, 4$



$d\ (\ l = 2\ )\ n = 3, 4, 5$



波動関数の角度依存

複素数は扱いにくいので実数の波動関数を用いる

$s : l = 0$

$$s = Y_{0,0} = \frac{1}{2\sqrt{\pi}}$$

```
load(draw);
d(f):=draw3d(color=red, spherical(if realpart(f)>0 then  f else 0,phi,0,2*%pi,theta,0,%pi),
             color=blue,spherical(if realpart(f)<0 then -f else 0,phi,0,2*%pi,theta,0,%pi),
             surface_hide=true);
define(s(theta,phi),Ylm(0,0,theta,phi));
d(s(theta,phi))$
```

s



波動関数の角度依存



$p : l = 1$

$$p_x = \frac{1}{\sqrt{2}}\left(Y_{1,-1} - Y_{1,1}\right) = \sqrt{\frac{3}{4\pi}}\,\frac{x}{r}$$

$$p_y = \frac{i}{\sqrt{2}}\left(Y_{1,-1} + Y_{1,1}\right) = \sqrt{\frac{3}{4\pi}}\,\frac{y}{r}$$

$$p_z = Y_{1,0} = \sqrt{\frac{3}{4\pi}}\,\frac{z}{r}$$

p_x



p_y



p_z



```
define(px(theta,phi),radcan((Ylm(1,-1,theta,phi)-Ylm(1,1,theta,phi))/sqrt(2)));
d(px(theta,phi))$
define(py(theta,phi),radcan((Ylm(1,-1,theta,phi)+Ylm(1,1,theta,phi)*%i)/sqrt(2)));
d(py(theta,phi))$
define(pz(theta,phi),radcan(Ylm(1,0,theta,phi)));
d(pz(theta,phi))$
```

波動関数の角度依存

$$d : l = 2$$

$$d_{z^2} = Y_{2,0} = \frac{1}{4}\sqrt{\frac{5}{\pi}}\left(3\frac{z^2}{r^2} - 1\right)$$

$$d_{x^2-y^2} = \frac{1}{\sqrt{2}}\left(Y_{2,2} + Y_{2,-2}\right) = \frac{1}{4}\sqrt{\frac{15}{\pi}}\,\frac{x^2-y^2}{r^2}$$

$$d_{xy} = \frac{1}{i\sqrt{2}}\left(Y_{2,2} - Y_{2,-2}\right) = \frac{1}{2}\sqrt{\frac{15}{\pi}}\,\frac{xy}{r^2}$$

$$d_{yz} = \frac{1}{i\sqrt{2}}\left(Y_{2,1} + Y_{2,-1}\right) = -\frac{1}{2}\sqrt{\frac{15}{\pi}}\,\frac{yz}{r^2}$$

$$d_{zx} = \frac{1}{\sqrt{2}}\left(Y_{2,1} - Y_{2,-1}\right) = -\frac{1}{2}\sqrt{\frac{15}{\pi}}\,\frac{zx}{r^2}$$

```
d(dz2(theta,phi))$
define(dx2y2(theta,phi),radcan((Ylm(2,2,theta,phi)+Ylm(2,-2,theta,phi))/sqrt(2)));
d(dx2y2(theta,phi))$
define(dxy(theta,phi),radcan((Ylm(2,2,theta,phi)-Ylm(2,-2,theta,phi))/(sqrt(2)*%i)));
d(dxy(theta,phi))$
define(dyz(theta,phi),radcan((Ylm(2,1,theta,phi)+Ylm(2,-1,theta,phi))/(sqrt(2)*%i)));
d(dyz(theta,phi))$
define(dzx(theta,phi),radcan((Ylm(2,1,theta,phi)-Ylm(2,-1,theta,phi))/sqrt(2)));
d(dzx(theta,phi))$
```

d_{z^2}



$d_{x^2-y^2}$



d_{xy}



d_{yz}



d_{zx}



波動関数をプロットする

```
load("orthopoly");
Rnl(n,l,r) := sqrt((2/(%a0*n))^3*(n-l-1)!/(2*n*((n+l)!)))*(2*r/(%a0*n))^l
              *exp(-r/(%a0*n))* gen_laguerre(n-l-1,2*l+1,2*r/(%a0*n));
Ylm(l,m,theta,phi) := spherical_harmonic(l,m,theta,phi);
%a0:1;
load(draw);
define(s(theta,phi),Ylm(0,0,theta,phi));
define(px(theta,phi),radcan((Ylm(1,-1,theta,phi)-Ylm(1,1,theta,phi))/sqrt(2)));
define(py(theta,phi),radcan((Ylm(1,-1,theta,phi)+Ylm(1,1,theta,phi)*%i)/sqrt(2)));
define(pz(theta,phi),radcan(Ylm(1,0,theta,phi)));
define(dz2(theta,phi),Ylm(2,0,theta,phi));
define(dx2y2(theta,phi),radcan((Ylm(2,2,theta,phi)+Ylm(2,-2,theta,phi))/sqrt(2)));
define(dxy(theta,phi),radcan((Ylm(2,2,theta,phi)-Ylm(2,-2,theta,phi))/(sqrt(2)*%i)));
define(dyz(theta,phi),radcan((Ylm(2,1,theta,phi)+Ylm(2,-1,theta,phi))/(sqrt(2)*%i)));
define(dzx(theta,phi),radcan((Ylm(2,1,theta,phi)-Ylm(2,-1,theta,phi))/sqrt(2)));
r:sqrt(x^2+y^2+z^2);
theta:acos(z/r);
phi:atan2(y,x);
define(psi_s(n,x,y,z)    ,Rnl(n,0,r)*s(theta,phi));
define(psi_px(n,x,y,z)   ,Rnl(n,1,r)*px(theta,phi));
define(psi_py(n,x,y,z)   ,Rnl(n,1,r)*py(theta,phi));
define(psi_pz(n,x,y,z)   ,Rnl(n,1,r)*pz(theta,phi));
define(psi_dz2(n,x,y,z)  ,Rnl(n,2,r)*dz2(theta,phi));
define(psi_dx2y2(n,x,y,z),Rnl(n,2,r)*dx2y2(theta,phi));
define(psi_dxy(n,x,y,z)  ,Rnl(n,2,r)*dxy(theta,phi));
define(psi_dyz(n,x,y,z)  ,Rnl(n,2,r)*dyz(theta,phi));
define(psi_zx(n,x,y,z)   ,Rnl(n,2,r)*dzx(theta,phi));
d(f):=draw3d(explicit(f,x,-30,30,z,-30,30),
       contour_levels=15,contour=both,surface_hide=true);
y:0.0001$
d(psi_s(1,x,y,z))$
d(psi_s(2,x,y,z))$
d(psi_px(2,x,y,z))$
d(psi_pz(2,x,y,z))$
d(psi_s(3,x,y,z))$
d(psi_px(3,x,y,z))$
d(psi_pz(3,x,y,z))$
d(psi_dz2(3,x,y,z))$
d(psi_dx2y2(3,x,y,z))$
d(psi_zx(3,x,y,z))$
```



分子結合

H_2^+分子を考える



座標の単位を Bohr 半径 (a_0=5.3x10^{-11}m)

エネルギーの単位を Hartree ($2Ry$ = 27.3eV)

にとる

$$H = -\frac{1}{2}\nabla^2 - \frac{1}{r_A} - \frac{1}{r_B} + \frac{1}{R}$$

$\Psi = c_A\phi_A + c_B\phi_B$ で固有関数と固有値を求める

$$\Psi_g = \frac{\phi_A + \phi_B}{\sqrt{2(1+S)}} \qquad E_g = \varepsilon + \frac{1}{R} - \frac{J+K}{1+S}$$

$$\Psi_u = \frac{\phi_A - \phi_B}{\sqrt{2(1+S)}} \qquad E_u = \varepsilon + \frac{1}{R} - \frac{J-K}{1-S}$$

ε: 1原子のエネルギー

$$J = \int \phi_A^* \frac{1}{r_B} \phi_A dV \qquad K = \int \phi_B^* \frac{1}{r_B} \phi_A dV \qquad S = \int \phi_A^* \phi_B dV$$

楕円座標

$$r_B\cos\theta_B - r_A\cos\theta_A = R$$
$$r_B\sin\theta_B = r_A\sin\theta_A$$
$$\Rightarrow \quad \cos\theta_A = \frac{r_B^2 - r_A^2 - R^2}{2Rr_A}, \qquad \cos\theta_B = \frac{r_B^2 - r_A^2 + R^2}{2Rr_B}$$

$$\Downarrow$$

$$r_A = \frac{\mu+\nu}{2}R, \qquad r_B = \frac{\mu-\nu}{2}R$$

$$\Rightarrow \quad \cos\theta_A = \frac{-\mu\nu-1}{\mu+\nu}, \qquad \sin\theta_A = \frac{\sqrt{\left(\mu^2-1\right)\left(1-\nu^2\right)}}{\mu+\nu}$$
$$\cos\theta_B = \frac{-\mu\nu+1}{\mu-\nu}, \qquad \sin\theta_B = \frac{\sqrt{\left(\mu^2-1\right)\left(1-\nu^2\right)}}{\mu-\nu}$$

$$\Downarrow \qquad 1 \le \mu,\ \ -1 \le \nu < 1,\ \ 0 \le \phi < 2\pi$$



A=(0, 0, *R*/2)

B=(0, 0, -*R*/2)

$$x = \frac{R}{2}\sqrt{\left(\mu^2-1\right)\left(1-\nu^2\right)}\cos\phi$$
$$y = \frac{R}{2}\sqrt{\left(\mu^2-1\right)\left(1-\nu^2\right)}\sin\phi$$
$$z = -\frac{R}{2}\mu\nu$$

$$\Rightarrow \quad \begin{vmatrix} \dfrac{\partial x}{\partial\mu}, & \dfrac{\partial x}{\partial\nu}, & \dfrac{\partial x}{\partial\phi} \\ \dfrac{\partial y}{\partial\mu}, & \dfrac{\partial y}{\partial\nu}, & \dfrac{\partial y}{\partial\phi} \\ \dfrac{\partial z}{\partial\mu}, & \dfrac{\partial z}{\partial\nu}, & \dfrac{\partial z}{\partial\phi} \end{vmatrix} = \frac{1}{8}R^3\left(\mu^2-\nu^2\right)$$

$$J = \int \phi_A^* \frac{1}{r_B}\phi_A dV = \frac{R^2}{4}\int_1^\infty d\mu \int_{-1}^1 d\nu \int_0^{2\pi} d\phi \left(\mu+\nu\right)\left|\phi_A\right|^2$$

$$K = \int \phi_B^* \frac{1}{r_B}\phi_A dV = \frac{R^2}{4}\int_1^\infty d\mu \int_{-1}^1 d\nu \int_0^{2\pi} d\phi \left(\mu+\nu\right)\phi_A\phi_B$$

$$S = \int \phi_A^*\phi_B dV = \frac{R^3}{8}\int_1^\infty d\mu \int_{-1}^1 d\nu \int_0^{2\pi} d\phi \left(\mu^2-\nu^2\right)\phi_A\phi_B$$

```
load("orthopoly");
Rnl(n,l,r) := sqrt((2/(%a0*n))^3*(n-l-1)!/(2*n*((n+l)!)))*(2*r/(%a0*n))^l
             *exp(-r/(%a0*n))* gen_laguerre(n-l-1,2*l+1,2*r/(%a0*n));
%a0:1;
define(psi(r),Rnl(1,0,r)/(2*sqrt(%pi)));
assume(R>0);
rA:(u+v)*R/2;
rB:(u-v)*R/2;
tA:acos((-u*v-1)/(u+v));
tB:acos((-u*v+1)/(u-v));
J(R):=R^2/4*2*%pi*integrate(integrate((u+v)*psi(rA)^2,          v,-1,1),u,1,inf);
K(R):=R^2/4*2*%pi*integrate(integrate((u+v)*psi(rA)*psi(rB),    v,-1,1),u,1,inf);
S(R):=R^3/8*2*%pi*integrate(integrate((u^2-v^2)*psi(rA)*psi(rB),v,-1,1),u,1,inf);
define(Eg(R),radcan(1/R-(J(R)+K(R))/(1+S(R))));
define(Eu(R),radcan(1/R-(J(R)-K(R))/(1-S(R))));
plot2d([Eg(R),Eu(R)],[R,0.2,8],[y,-0.2,0.4]);
load(newton);
Rmin:newton(diff(Eg(x),x),2);
Emin:Eg(Rmin);
```

$$Eg(R) := -\frac{(2R^2-3)\%e^R - 3R - 3}{3R\%e^{2R} + (R^3+3R^2+3R)\%e^R}$$

$$Eu(R) := \frac{(2R^2-3)\%e^R + 3R + 3}{3R\%e^{2R} + (-R^3-3R^2-3R)\%e^R}$$

2.492830410358391b0

−6.483099237080779b-2



		実際の値
ボンド長 : 2.49	→ 0.13 nm	0.11 nm
結合エネルギー : 0.0648	1.76 eV	2.78 eV


エネルギー
反結合軌道
結合軌道
原子1
分子
原子2
原子軌道
分子軌道
結合軌道
反結合軌道
波動関数
電子の存在確率


s 軌道
σs
結合軌道
σs*
反結合軌道
pz 軌道
σp
結合軌道
σp*
反結合軌道
px 軌道
πp
結合軌道
πp*
反結合軌道
x
z

混成軌道

C : $1s^2 2s^2 2p^2$





昇位





原子軌道のエネルギー
としては損

結合数が増えることにより
エネルギーが得

sp^3 混成軌道

$$\psi_1 = \frac{1}{2}\left(\phi_{2s} + \phi_{2p_x} + \phi_{2p_y} + \phi_{2p_z}\right)$$

$$\psi_2 = \frac{1}{2}\left(\phi_{2s} + \phi_{2p_x} - \phi_{2p_y} - \phi_{2p_z}\right)$$

$$\psi_3 = \frac{1}{2}\left(\phi_{2s} - \phi_{2p_x} + \phi_{2p_y} - \phi_{2p_z}\right)$$

$$\psi_4 = \frac{1}{2}\left(\phi_{2s} - \phi_{2p_x} - \phi_{2p_y} + \phi_{2p_z}\right)$$





ダイヤモンド構造



```
sp3a(x,y,z):=(psi_s(2,x,y,z)+psi_px(2,x,y,z)+psi_py(2,x,y,z)+psi_pz(2,x,y,z))/2;
sp3b(x,y,z):=(psi_s(2,x,y,z)+psi_px(2,x,y,z)-psi_py(2,x,y,z)-psi_pz(2,x,y,z))/2;
sp3c(x,y,z):=(psi_s(2,x,y,z)-psi_px(2,x,y,z)+psi_py(2,x,y,z)-psi_pz(2,x,y,z))/2;
sp3d(x,y,z):=(psi_s(2,x,y,z)-psi_px(2,x,y,z)-psi_py(2,x,y,z)+psi_pz(2,x,y,z))/2;
y:0.0001;
d(f):=draw3d(explicit(f,x,-10,10,z,-10,10),
      contour_levels=15,contour=both,surface_hide=true);
d(sp3a(x,y,z))$
d(sp3b(x,y,z))$
d(sp3b(x,y,z))$
d(sp3b(x,y,z))$
```